
Travaux Pratiques n°1 :

Elements finis

Seatech - MOCA

Réalisé par : Thomas DE GUZMAN DE SAINT NICOLAS
Enseignant ecadrant : Mr J. Schneider, enseignant
Date de rendu : 20 décembre 2025

Table des matières

1	Introduction	2
2	Présentation du problème	3
3	Partie théorique - Méthode des éléments finis	5
3.1	Formulation forte du problème	5
3.2	Passage à la formulation faible : méthode du test	5
3.3	Discretisation du domaine	6
3.4	Espace d'approximation et fonctions de forme	6
3.5	Écriture élément par élément	7
3.6	Définition et calcul de la matrice élémentaire	9
3.7	Assemblage global (méthode du cours)	10
3.8	Conditions aux limites de Dirichlet	11
3.9	Système linéaire final	11
4	Étude de notre problème	12
4.1	Rappel du problème	12
4.2	Domaine de calcul et conditions aux limites	12
4.3	Matrice élémentaire	13
4.4	Assemblage de la matrice globale	13
4.4.1	Numérotation des nœuds et connectivité	13
4.4.2	Contribution de l'élément K_1	14
4.4.3	Contribution des autres éléments	15
4.4.4	Matrice de rigidité globale	16
4.5	Champ de solution numérique	17
4.6	Influence du maillage	17
4.7	Comportement au voisinage des coins	18
5	Étude de l'erreur	19
5.1	Définition de l'erreur	19
5.2	Choix d'une solution de référence	19
5.3	Calcul numérique de l'erreur	19
5.4	Convergence en fonction du pas de discrétisation	20
5.5	Analyse et conclusion sur l'étude de l'erreur	21
6	Conclusion	22

1 Introduction

La méthode des éléments finis est largement utilisée pour approcher numériquement la solution d'équations aux dérivées partielles. Dans ce TP, nous l'appliquons à un problème elliptique défini sur le domaine carré $[0, 10] \times [0, 10]$.

Le domaine est discrétisé à l'aide d'un maillage uniforme de pas $h = \frac{10}{n-1}$, ce qui conduit à n^2 nœuds numérotés ligne par ligne. Chaque cellule du maillage est modélisée par un élément quadrilatéral associé à une matrice élémentaire de rigidité 4×4 fournie par l'énoncé.

L'assemblage de l'ensemble des matrices élémentaires permet de former la matrice globale du problème. Les conditions aux limites sont ensuite imposées par une méthode de pénalisation, qui consiste à modifier les lignes correspondant aux nœuds du bord afin de fixer la valeur de la solution sur ces points.

Une fois la matrice globale construite et les conditions au bord appliquées, le système linéaire obtenu est résolu numériquement. Les calculs sont répétés pour plusieurs tailles de maillage ($n = 10, 20, 100$) afin d'étudier l'influence de la discrétisation sur la solution obtenue.

Ce rapport présente donc les étapes essentielles de la démarche : construction du maillage, connectivité, assemblage, imposition des conditions au bord, résolution et analyse des résultats.

2 Présentation du problème

On cherche à résoudre un problème elliptique posé sur le domaine carré


$$\Omega = [0, 10] \times [0, 10].$$

Le problème consiste à déterminer une fonction $\lambda(x, y)$ satisfaisant le système linéaire issu de la méthode des éléments finis. Les conditions au bord imposent une valeur donnée de la solution sur tout le contour, en particulier :

$$\lambda(x, 10) = 100, \quad \lambda(x, 0) = 0, \quad \lambda(0, y) = 0, \quad \lambda(10, y) = 0.$$

Le domaine est illustré ci-dessous :

$$y = 10 \text{ (CL} = 100\text{)}$$



L'objectif est de discrétiser ce domaine, construire la matrice globale d'éléments finis et résoudre :

$$M\lambda = b,$$

où :

- M est la matrice de rigidité assemblée
- b le second membre défini par les conditions au bord.

Maillage du domaine

Le domaine est discrétisé à l'aide d'un maillage uniforme comportant n nœuds dans chaque direction. Le pas de discrétisation est

$$h = \frac{10}{n-1}.$$

Coordonnées et numérotation


Les nœuds sont numérotés **ligne par ligne**, de gauche à droite et du bas vers le haut. La coordonnée du nœud (i, j) est :

$$x_i = (i - 1)h, \quad y_j = (j - 1)h.$$

La numérotation globale des noeuds (bord inclus) est donnée par :

$$\text{Id}(i, j) = i + (j - 1)n, \quad i, j = 1, \dots, n.$$

Voici un schéma pour $n = 4$ (3×3 éléments) :



Connectivité des éléments

Un élément quadrilatéral situé entre les nœuds (i, j) et $(i + 1, j + 1)$ possède quatre sommets numérotés en sens trigonométrique :

$$\{K_1, K_2, K_3, K_4\} = \{\text{Id}(i, j), \text{Id}(i+1, j), \text{Id}(i+1, j+1), \text{Id}(i, j+1)\}.$$

Ainsi, pour $n = 3$ (9 noeuds, 4 éléments). Les noeuds sont numérotés $\begin{smallmatrix} 7 & 8 & 9 \\ 4 & 5 & 6 \\ 1 & 2 & 3 \end{smallmatrix}$, et l'on obtient les 4 éléments quadrilatéraux :

$$\begin{aligned} K_1 &= (1, 2, 5, 4), & K_2 &= (2, 3, 6, 5), \\ K_3 &= (4, 5, 8, 7), & K_4 &= (5, 6, 9, 8). \end{aligned}$$

Élément	Connectivité $\{K_1, K_2, K_3, K_4\}$ (sens trigo)
K_1	(1, 2, 5, 4)
K_2	(2, 3, 6, 5)
K_3	(4, 5, 8, 7)
K_4	(5, 6, 9, 8)

3 Partie théorique - Méthode des éléments finis

Cette section présente l'ensemble des outils mathématiques nécessaires pour établir la formulation variationnelle puis discrète du problème elliptique étudié.

Elle suit les étapes classiques de la méthode des éléments finis : formulation forte, formulation faible, discrétisation du domaine, définition des fonctions de forme, calcul de la matrice élémentaire, assemblage global et traitement des conditions aux limites.

3.1 Formulation forte du problème

On cherche une fonction $u : \Omega \rightarrow \mathbb{R}$ solution de l'équation de Laplace

$$-\Delta u = 0 \quad \text{dans } \Omega.$$

avec conditions de Dirichlet imposées sur le bord :

$$u = g \quad \text{sur } \partial\Omega.$$

Dans notre TP :

$$u = \begin{cases} 100 & \text{sur le bord supérieur } \{(x, 10)\}, \\ 0 & \text{ailleurs sur } \partial\Omega. \end{cases}$$

L'objectif est de construire une approximation numérique u_h de u par la méthode des éléments finis.

3.2 Passage à la formulation faible : méthode du test

On multiplie l'équation par une fonction test v appartenant à :

$$V_0 = H_0^1(\Omega) = \{v \in H^1(\Omega) : v|_{\partial\Omega} = 0\}.$$

En intégrant sur Ω on obtient :

$$\int_{\Omega} (-\Delta u) v \, dx = 0.$$

En utilisant l'identité de Green (intégration par parties) on a :

$$\int_{\Omega} -(\Delta u) v = \int_{\Omega} \nabla u \cdot \nabla v \, dx - \int_{\partial\Omega} \frac{\partial u}{\partial n} v \, ds = 0.$$

Comme $v = 0$ sur $\partial\Omega$, le terme de bord est nul. On obtient la formulation faible :

$$\int_{\Omega} -(\Delta u)v \, dx = \int_{\Omega} \nabla u \cdot \nabla v \, dx - \int_{\partial\Omega} \frac{\partial u}{\partial n} v \, ds = 0.$$

$$\implies \boxed{\int_{\Omega} \nabla u \cdot \nabla v \, dx = 0 \quad \forall v \in V_0.}$$

On définit ensuite la forme bilinéaire

$$a(u, v) = \int_{\Omega} \nabla u \cdot \nabla v \, dx.$$

Le problème variationnel devient alors :

$$\boxed{\text{Trouver } u \in H_g^1(\Omega) \text{ tel que } a(u, v) = 0, \quad \forall v \in V_0.}$$

3.3 Discrétisation du domaine

On subdivise le domaine Ω en un maillage régulier formé de $(n-1)^2$ éléments quadrangulaires et de pas :

$$h = \frac{10}{n-1},$$

Les nœuds sont numérotés ligne par ligne :

$$\text{Id}(i, j) = i + (j-1)n.$$

Chaque élément E_k possède quatre sommets globaux notés

$$A_1^{(k)}, A_2^{(k)}, A_3^{(k)}, A_4^{(k)}.$$

3.4 Espace d'approximation et fonctions de forme

Pour construire l'approximation éléments finis, on introduit un espace discret V_h constitué de fonctions définies à partir du maillage. On associe ensuite une fonction de forme ϕ_i à chaque nœud du maillage. Le nombre total de nœuds étant $N_n = n^2$, on définit :

$$V_h = \text{Span}\{\phi_1, \phi_2, \dots, \phi_{N_n}\} \subset H^1(\Omega).$$

Chaque fonction de forme satisfait la propriété nodale :

$$\phi_i(P_j) = \delta_{ij} = \begin{cases} 1 & \text{si } i = j, \\ 0 & \text{si } i \neq j. \end{cases}$$

Ainsi, le coefficient u_i représente directement la valeur approchée de la solution au nœud P_i .

Sur un élément carré $E_k = [0, h] \times [0, h]$, on utilise des fonctions de forme bilinéaires. Elles sont définies localement par :

$$\begin{aligned}\phi_1^{(k)}(x, y) &= (1 - \frac{x}{h})(1 - \frac{y}{h}), \\ \phi_2^{(k)}(x, y) &= \frac{x}{h}(1 - \frac{y}{h}), \\ \phi_3^{(k)}(x, y) &= \frac{x}{h} \frac{y}{h}, \\ \phi_4^{(k)}(x, y) &= (1 - \frac{x}{h}) \frac{y}{h}.\end{aligned}$$

Chacune de ces fonctions vaut 1 sur un sommet de E_k et 0 sur les trois autres, ce qui garantit une interpolation locale cohérente.

La solution approchée s'écrit alors sous la forme :

$$u_h(x, y) = \sum_{i=1}^{N_n} u_i \phi_i(x, y),$$

C'est-à-dire une combinaison linéaire de toutes les fonctions de forme, pondérée par les valeurs inconnues u_i aux nœuds.

3.5 Écriture élément par élément

La méthode des éléments finis s'appuie sur l'idée que l'intégrale sur tout le domaine peut être décomposée en une somme d'intégrales sur chacun des éléments du maillage. Ainsi, pour toute intégrale définie sur Ω , on a :

$$\int_{\Omega} (\cdot) dx = \sum_{k=1}^{N_e} \int_{E_k} (\cdot) dx.$$

En particulier, en prenant la fonction test $v = \phi_p$ (fonction de forme associée au nœud global P_p), la formulation faible s'écrit :

$$\int_{\Omega} \nabla u_h \cdot \nabla \phi_p dx = 0,$$

et donc :

$$\sum_{k=1}^{N_e} \int_{E_k} \nabla u_h \cdot \nabla \phi_p dx = 0.$$

On fait ensuite la décomposition locale de u_h .

En effet, sur un élément E_k , la solution approchée u_h ne dépend que des quatre nœuds de cet élément.

On peut donc l'écrire sous la forme :

$$u_h|_{E_k} = \sum_{j=1}^4 u_{A_j^{(k)}} \phi_j^{(k)},$$

où :

- $A_j^{(k)}$ est le numéro global du j -ème sommet de l'élément E_k ,
- $\phi_j^{(k)}$ est la fonction de forme locale associée à ce sommet.

En dérivant cette expression, on obtient le gradient local :

$$\nabla u_h|_{E_k} = \sum_{j=1}^4 u_{A_j^{(k)}} \nabla \phi_j^{(k)}.$$

En remplaçant ∇u_h dans l'intégrale sur E_k , on obtient :

$$\int_{E_k} \nabla u_h \cdot \nabla \phi_p dx = \int_{E_k} \left(\sum_{j=1}^4 u_{A_j^{(k)}} \nabla \phi_j^{(k)} \right) \cdot \nabla \phi_p dx.$$

Par linéarité de l'intégrale on a alors :

$$\int_{E_k} \nabla u_h \cdot \nabla \phi_p dx = \sum_{j=1}^4 u_{A_j^{(k)}} \int_{E_k} \nabla \phi_j^{(k)} \cdot \nabla \phi_p dx.$$

En reportant dans la somme sur tous les éléments, on obtient :

$$\sum_{k=1}^{N_e} \sum_{j=1}^4 u_{A_j^{(k)}} \int_{E_k} \nabla \phi_j^{(k)} \cdot \nabla \phi_p dx = 0.$$

Cette écriture permet d'identifier directement les coefficients de la matrice élémentaire :

$$A_{ij}^{(k)} = \int_{E_k} \nabla \phi_i^{(k)} \cdot \nabla \phi_j^{(k)} dx.$$

3.6 Définition et calcul de la matrice élémentaire

Pour tout élément $E_k = [0, h] \times [0, h]$, la matrice élémentaire est définie par :

$$A_{ij}^{(k)} = \int_{E_k} \nabla \phi_i^{(k)}(x, y) \cdot \nabla \phi_j^{(k)}(x, y) dx dy.$$

Les fonctions de forme locales sont :

$$\phi_1 = (1 - \frac{x}{h})(1 - \frac{y}{h}), \quad \phi_2 = \frac{x}{h}(1 - \frac{y}{h}), \quad \phi_3 = \frac{x}{h}\frac{y}{h}, \quad \phi_4 = (1 - \frac{x}{h})\frac{y}{h}.$$

En dérivant ces expressions par rapport à x et y, on obtient les gradients $\nabla \phi_i$,

$$\partial_x \phi_1 = -\frac{1}{h} + \frac{y}{h^2}, \quad \text{et} \quad \partial_y \phi_1 = -\frac{1}{h} + \frac{x}{h^2}, \quad \Rightarrow \quad \nabla \phi_1(x, y) = \left(-\frac{1}{h} + \frac{y}{h^2}, -\frac{1}{h} + \frac{x}{h^2} \right)$$

$$\partial_x \phi_2 = \frac{1}{h} - \frac{y}{h^2}, \quad \text{et} \quad \partial_y \phi_2 = -\frac{x}{h^2}, \quad \Rightarrow \quad \nabla \phi_2(x, y) = \left(\frac{1}{h} - \frac{y}{h^2}, -\frac{x}{h^2} \right)$$

$$\partial_x \phi_3 = \frac{y}{h^2}, \quad \text{et} \quad \partial_y \phi_3 = \frac{x}{h^2}, \quad \Rightarrow \quad \nabla \phi_3(x, y) = \left(\frac{y}{h^2}, \frac{x}{h^2} \right)$$

$$\partial_x \phi_4 = -\frac{y}{h^2}, \quad \text{et} \quad \partial_y \phi_4 = \frac{1}{h} - \frac{x}{h^2}, \quad \Rightarrow \quad \nabla \phi_4(x, y) = \left(-\frac{y}{h^2}, \frac{1}{h} - \frac{x}{h^2} \right)$$

Puis on obtient ensuite les produits scalaires $\nabla \phi_i \cdot \nabla \phi_j$:

$$\nabla \phi_i \cdot \nabla \phi_j = \begin{pmatrix} \nabla \phi_1 \cdot \nabla \phi_1 & \nabla \phi_1 \cdot \nabla \phi_2 & \nabla \phi_1 \cdot \nabla \phi_3 & \nabla \phi_1 \cdot \nabla \phi_4 \\ \nabla \phi_2 \cdot \nabla \phi_1 & \nabla \phi_2 \cdot \nabla \phi_2 & \nabla \phi_2 \cdot \nabla \phi_3 & \nabla \phi_2 \cdot \nabla \phi_4 \\ \nabla \phi_3 \cdot \nabla \phi_1 & \nabla \phi_3 \cdot \nabla \phi_2 & \nabla \phi_3 \cdot \nabla \phi_3 & \nabla \phi_3 \cdot \nabla \phi_4 \\ \nabla \phi_4 \cdot \nabla \phi_1 & \nabla \phi_4 \cdot \nabla \phi_2 & \nabla \phi_4 \cdot \nabla \phi_3 & \nabla \phi_4 \cdot \nabla \phi_4 \end{pmatrix}.$$

Chaque terme de $A_{ij}^{(k)}$ résulte alors de l'intégration suivante :

$$A_{ij}^{(k)} = \int_0^h \int_0^h (\partial_x \phi_i \partial_x \phi_j + \partial_y \phi_i \partial_y \phi_j) dx dy.$$

Les calculs explicites conduisent alors à :

$$A^{(k)} = \frac{1}{6} \begin{pmatrix} 4 & -1 & -2 & -1 \\ -1 & 4 & -1 & -2 \\ -2 & -1 & 4 & -1 \\ -1 & -2 & -1 & 4 \end{pmatrix}.$$

Or, le maillage est régulier et que les fonctions de forme sont identiques pour chaque élément.

Donc la matrice élémentaire est la même pour tous les éléments.

On a alors :

$$A^{(k)} = A^{(\ell)} \quad \forall k, \ell.$$

3.7 Assemblage global (méthode du cours)

Une fois la matrice élémentaire $A^{(k)}$ calculée pour un élément E_k , il reste à construire la matrice globale $A \in \mathbb{R}^{N_n \times N_n}$ définie sur l'ensemble du maillage. Comme dans les exercices de 1D vus en cours, cette étape repose sur la correspondance entre :

- les indices locaux $i, j = 1, \dots, 4$ des sommets de l'élément E_k ,
- les indices globaux $A_i^{(k)}, A_j^{(k)}$ des nœuds du maillage.

Pour chaque élément E_k , on dispose d'un tableau de connectivité qui donne explicitement les quatre nœuds globaux :

$$(A_1^{(k)}, A_2^{(k)}, A_3^{(k)}, A_4^{(k)}).$$

L'assemblage consiste à ajouter les contributions élémentaires dans les bonnes cases de la matrice globale. Autrement dit, chaque coefficient local $A_{ij}^{(k)}$ doit être ajouté au coefficient global qui correspond aux nœuds globaux $A_i^{(k)}$ et $A_j^{(k)}$:

$$A_{pq} += A_{ij}^{(k)} \quad \text{dès que } p = A_i^{(k)} \text{ et } q = A_j^{(k)}.$$

Ainsi, pour chaque élément E_k :

Pour $i, j = 1, \dots, 4$, $A(A_i^{(k)}, A_j^{(k)}) \leftarrow A(A_i^{(k)}, A_j^{(k)}) + A_{ij}^{(k)}$.

Ce processus se répète pour tous les éléments du maillage. Comme plusieurs éléments adjacents partagent des nœuds communs, leurs contributions s'additionnent automatiquement dans la matrice globale.

On obtient ainsi une matrice globale :

$$A \in \mathbb{R}^{N_n \times N_n},$$

qui est :

- symétrique (car $A^{(k)}$ est symétrique),
- définie positive,
- très creuse (chaque ligne ne contient que quelques non-zéros),
- structurée suivant la topologie du maillage.

Cette matrice est ensuite modifiée pour tenir compte des conditions aux limites.

3.8 Conditions aux limites de Dirichlet

On impose :

$$u_i = g_i, \quad i \in \partial\Omega.$$

Comme dans le TD, on utilise la pénalisation :

$$A_{ii} \leftarrow 10^6, \quad b_i \leftarrow 10^6 g_i.$$

Tous les autres coefficients de la ligne i sont mis à zéro.

Le choix d'un coefficient de pénalisation très grand permet de forcer la valeur nodale imposée tout en conservant un système linéaire de même taille. Cette méthode est cohérente tant que le terme de pénalisation reste très supérieur aux coefficients naturels de la matrice de rigidité.

3.9 Système linéaire final

On obtient :

$$\boxed{A u = b.}$$

La solution approchée est :

$$u_h(x, y) = \sum_{i=1}^{N_n} u_i \phi_i(x, y).$$

4 Étude de notre problème

Dans cette section, on analyse la solution numérique obtenue par la méthode des éléments finis appliquée au problème elliptique étudié. Les résultats sont issus du programme de calcul développé en FORTRAN et sont visualisés à l'aide du logiciel VISIT.

L'objectif est de mettre en évidence le comportement spatial de la solution, d'évaluer l'influence du raffinement du maillage et d'interpréter les résultats à la lumière des propriétés mathématiques de l'équation de Laplace et de la théorie des éléments finis. Afin de faciliter l'analyse des structures matricielles et des champs numériques, on se place dans un premier temps dans le cas d'un maillage volontairement grossier ($n = 3$).

4.1 Rappel du problème

On considère l'équation de Laplace homogène


$$-\Delta u = 0 \quad \text{dans } \Omega = [0, 10] \times [0, 10],$$

soumise à des conditions aux limites de Dirichlet imposant une valeur constante sur le bord supérieur du domaine et une valeur nulle sur les trois autres bords.

Ce type de configuration correspond à un problème de diffusion stationnaire sans source interne, dans lequel la fonction inconnue peut être interprétée comme un champ de température à l'équilibre.

4.2 Domaine de calcul et conditions aux limites

La géométrie du domaine et les conditions aux limites imposées sont représentées sur la figure suivante :



La valeur imposée sur le bord supérieur diffuse progressivement à l'intérieur du domaine, tandis que les autres bords maintenus à zéro agissent comme des dissipateurs.

4.3 Matrice élémentaire

La matrice élémentaire associée à chaque élément quadrilatéral est fournie par l'énoncé et correspond à un élément bilinéaire de type Q1. Elle est donnée par :

$$K_e = \frac{1}{6} \begin{pmatrix} 4 & -1 & -2 & -1 \\ -1 & 4 & -1 & -2 \\ -2 & -1 & 4 & -1 \\ -1 & -2 & -1 & 4 \end{pmatrix}.$$

Cette matrice représente la contribution locale d'un élément au terme de diffusion issu de l'équation de Laplace. Elle est symétrique et définie positive, ce qui garantit la stabilité numérique du schéma éléments finis.

Dans le cas présent, le maillage étant régulier et tous les éléments ayant la même géométrie, la matrice élémentaire est identique pour chaque élément du domaine.

4.4 Assemblage de la matrice globale

On décrit dans cette section la procédure d'assemblage de la matrice de rigidité globale issue de la méthode des éléments finis. Afin d'illustrer clairement le mécanisme, on se place dans le cas d'un maillage volontairement grossier comportant $n = 3$ nœuds par direction, soit 9 nœuds et 4 éléments quadrilatéraux de type Q1.

4.4.1 Numérotation des nœuds et connectivité

Les nœuds sont numérotés de manière lexicographique, ligne par ligne, de gauche à droite et de bas en haut. Les éléments sont définis par les connectivités suivantes :

$$K_1 : (1, 2, 5, 4),$$

$$K_2 : (2, 3, 6, 5),$$

$$K_3 : (4, 5, 8, 7),$$

$$K_4 : (5, 6, 9, 8).$$

Chaque élément contribue à la matrice globale par l'intermédiaire de la matrice élémentaire

K_e , associée aux éléments quadrilatéraux bilinéaires Q1 :

$$K_e = \frac{1}{6} \begin{pmatrix} 4 & -1 & -2 & -1 \\ -1 & 4 & -1 & -2 \\ -2 & -1 & 4 & -1 \\ -1 & -2 & -1 & 4 \end{pmatrix}.$$

Les coefficients de cette matrice proviennent de l'expression générale

$$(K_e)_{ij} = \int_{K_e} \nabla \varphi_i \cdot \nabla \varphi_j d\Omega,$$

où φ_i désigne les fonctions de forme locales associées aux sommets de l'élément.

4.4.2 Contribution de l'élément K_1

L'élément K_1 dépend des nœuds globaux (1, 2, 5, 4). Sa contribution à la matrice globale de taille 9×9 est obtenue en plaçant les coefficients de K_e aux lignes et colonnes correspondant à ces indices globaux :

$$K^{(1)} = \frac{1}{6} \begin{pmatrix} 4 & -1 & 0 & -1 & -2 & 0 & 0 & 0 & 0 \\ -1 & 4 & 0 & -2 & -1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ -1 & -2 & 0 & 4 & -1 & 0 & 0 & 0 & 0 \\ -2 & -1 & 0 & -1 & 4 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{pmatrix}.$$

Les lignes et colonnes associées aux nœuds ne participant pas à l'élément sont nulles.

4.4.3 Contribution des autres éléments

De manière analogue, l'élément K_2 , connecté aux nœuds (2, 3, 6, 5), fournit la contribution :

$$K^{(2)} = \frac{1}{6} \begin{pmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 4 & -1 & 0 & -1 & -2 & 0 & 0 & 0 \\ 0 & -1 & 4 & 0 & -2 & -1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & -1 & -2 & 0 & 4 & -1 & 0 & 0 & 0 \\ 0 & -2 & -1 & 0 & -1 & 4 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{pmatrix}.$$

L'élément K_3 est connecté aux nœuds globaux (4, 5, 8, 7). Sa contribution à la matrice globale s'écrit :

$$K^{(3)} = \frac{1}{6} \begin{pmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 4 & -1 & 0 & -1 & -2 & 0 \\ 0 & 0 & 0 & -1 & 4 & 0 & -2 & -1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & -1 & -2 & 0 & 4 & -1 & 0 \\ 0 & 0 & 0 & -2 & -1 & 0 & -1 & 4 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{pmatrix}.$$

De manière analogue, l'élément K_4 , associé aux nœuds globaux (5, 6, 9, 8), contribue sous la forme :

$$K^{(4)} = \frac{1}{6} \begin{pmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 4 & -1 & 0 & -1 & -2 \\ 0 & 0 & 0 & 0 & -1 & 4 & 0 & -2 & -1 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & -1 & -2 & 0 & 4 & -1 \\ 0 & 0 & 0 & 0 & -2 & -1 & 0 & -1 & 4 \end{pmatrix}.$$

Ces deux contributions complètent l'assemblage pour le cas $n = 3$. La matrice globale est obtenue par sommation des quatre contributions élémentaires :

$$K = K^{(1)} + K^{(2)} + K^{(3)} + K^{(4)}.$$

4.4.4 Matrice de rigidité globale

L'assemblage de la matrice globale consiste à ajouter, pour chaque élément du maillage, les coefficients de la matrice élémentaire K_e aux positions globales correspondant aux quatre nœuds de cet élément.

Cette opération est réalisée en suivant la connectivité du maillage : chaque coefficient local est reporté dans la matrice globale aux indices associés aux nœuds globaux de l'élément.




FIGURE 1 – Influence du raffinement du maillage sur la solution numérique.


On obtient ainsi la matrice :

$$M = \frac{1}{6} \begin{bmatrix} 4 & -1 & 0 & -1 & 2 & 0 & 0 & 0 & 0 \\ -1 & 8 & -1 & -2 & -2 & -2 & 0 & 0 & 0 \\ 0 & -1 & 4 & 0 & -2 & -1 & 0 & 0 & 0 \\ -1 & -2 & 0 & 8 & -2 & 0 & -1 & -2 & 0 \\ -2 & -2 & -2 & -2 & 16 & -2 & -2 & -2 & -2 \\ 0 & -2 & -1 & 0 & -2 & 8 & 0 & -2 & -1 \\ 0 & 0 & 0 & -1 & -2 & 0 & 4 & -1 & 0 \\ 0 & 0 & 0 & -2 & -2 & -2 & -1 & 8 & -1 \\ 0 & 0 & 0 & 0 & -2 & -1 & 0 & -1 & 4 \end{bmatrix}$$

Cette matrice est symétrique, définie positive et creuse. Sa structure reflète la connectivité locale du maillage, chaque nœud n'interagissant qu'avec ses voisins immédiats, propriété essentielle pour l'efficacité des méthodes numériques.

4.5 Champ de solution numérique

La figure ci-dessus présente le champ de la solution numérique obtenu par éléments finis, visualisé sous forme de carte de couleurs à l'aide de VISIT.



On observe que la solution atteint sa valeur maximale sur le bord supérieur, conformément aux conditions aux limites imposées, puis décroît progressivement vers les autres bords du domaine. La solution est continue et régulière à l'intérieur du domaine, ce qui est caractéristique d'un problème elliptique stationnaire.

FIGURE 2 – Champ de la solution numérique pour un maillage de taille $n = 3$.

4.6 Influence du maillage

Afin d'étudier l'influence du raffinement du maillage, les calculs ont été réalisés pour plusieurs tailles de maillage :

$$n = 10, \quad 20, \quad 100.$$




FIGURE 3 – Influence du raffinement du maillage sur la solution numérique.

Lorsque le maillage est grossier, la solution est correctement capturée dans ses grandes lignes, mais certaines zones, notamment au voisinage des coins, sont moins bien résolues. Le raffinement du maillage permet une meilleure description des gradients et une amélioration progressive de la précision de la solution.

4.7 Comportement au voisinage des coins

Les coins du domaine correspondent à des zones où les conditions aux limites changent brutalement de valeur. Ces discontinuités induisent des gradients plus importants, qui sont plus difficiles à capturer avec un maillage grossier.

Les visualisations montrent que le raffinement du maillage améliore la résolution de ces zones, ce qui est cohérent avec la théorie des éléments finis.

5 Étude de l'erreur

L'objectif de cette section est d'évaluer quantitativement la précision de la solution éléments finis obtenue, et d'étudier la convergence de la méthode lorsque le maillage est raffiné.

On s'intéresse à l'évolution de l'erreur en fonction du pas de discrétisation h , ainsi qu'à la cohérence du comportement observé avec la théorie des éléments finis pour un problème elliptique.

5.1 Définition de l'erreur

Dans le cadre de la méthode des éléments finis, l'erreur est définie comme la différence entre la solution exacte u du problème continu et la solution approchée u_h obtenue sur le maillage.

Lorsque la solution exacte est connue, on peut définir l'erreur dans différentes normes. Dans ce travail, on s'intéresse à la norme L^2 , définie par :

$$\|u - u_h\|_{L^2(\Omega)} = \left(\int_{\Omega} (u(x, y) - u_h(x, y))^2 dx dy \right)^{1/2}.$$

5.2 Choix d'une solution de référence

Dans le cas présent, le problème de Laplace avec conditions aux limites de Dirichlet non homogènes ne possède pas de solution analytique simple exploitable sur l'ensemble du domaine. Afin d'évaluer quantitativement l'erreur de la solution éléments finis, une approche classique en calcul numérique est adoptée.

Une solution numérique de référence u_{ref} , calculée sur un maillage très fin (par exemple $n = 100$ ou plus), est utilisée comme approximation de la solution exacte. Cette solution est supposée suffisamment précise pour servir de base de comparaison lors de l'étude de la convergence globale.

Dans une seconde approche, une solution analytique issue d'un développement en série de Fourier est également exploitée. Cette solution analytique permet d'analyser finement la répartition spatiale de l'erreur et de valider qualitativement la solution numérique, en particulier au voisinage des singularités induites par les conditions aux limites.

5.3 Calcul numérique de l'erreur

Pour chaque maillage de pas h , l'erreur est évaluée en comparant la solution u_h à la solution de référence u_{ref} .

Dans le code Fortran, cette erreur est calculée de manière discrète sous la forme :


$$\|u_h - u_{\text{ref}}\|_{L^2} \approx \left(\sum_{i=1}^{N_n} (u_i - u_{\text{ref},i})^2 h^2 \right)^{1/2},$$

où u_i et $u_{\text{ref},i}$ désignent les valeurs nodales de la solution approchée et de la solution de référence.

Le facteur h^2 correspond à l'aire associée à chaque nœud du maillage uniforme.

5.4 Convergence en fonction du pas de discrétisation

Les calculs ont été réalisés pour plusieurs tailles de maillage, correspondant à différentes valeurs du pas $h = \frac{10}{n-1}$.



Les cartes d'erreur locale obtenues pour différents raffinements de maillage sont présentées sur la Figure 5.4. Pour des maillages grossiers ($n = 3$ et $n = 10$), l'erreur est répartie de manière relativement homogène sur l'ensemble du domaine. Ce comportement traduit une approximation encore insuffisante de la solution, le nombre limité de degrés de liberté ne permettant pas de capturer correctement les gradients associés au phénomène de diffusion.

Lorsque le maillage est raffiné ($n \geq 20$), l'erreur se localise progressivement et se concentre principalement au voisinage des coins supérieurs du domaine. Cette localisation est caractéristique de la présence de singularités induites par la discontinuité des conditions aux limites, tandis que l'erreur devient négligeable dans le reste du domaine.

Une représentation de l'erreur globale en norme L^2 en fonction du pas de discrétisation h , en échelle logarithmique, permettrait d'identifier directement l'ordre de convergence de la méthode par la pente de la courbe obtenue.

5.5 Analyse et conclusion sur l'étude de l'erreur

Les résultats numériques montrent que l'erreur décroît lorsque le maillage est raffiné, ce qui confirme la convergence de la méthode des éléments finis. La représentation de l'erreur en échelle logarithmique met en évidence une décroissance quasi linéaire, correspondant à un ordre de convergence proche de 1.

En l'absence de singularités, l'ordre de convergence théorique en norme L^2 pour des éléments quadrilatéraux bilinéaires (Q1) est de l'ordre de 2. Toutefois, la présence de discontinuités dans les conditions aux limites aux coins du domaine réduit la régularité de la solution exacte. Cette perte de régularité limite l'ordre de convergence effectivement observé et explique l'écart avec l'ordre théorique optimal.

Cette étude met ainsi en évidence le lien étroit entre la régularité de la solution, la nature des conditions aux limites et les performances de la méthode des éléments finis. Les résultats obtenus sont en accord avec les prédictions théoriques pour les problèmes elliptiques présentant des singularités locales.

6 Conclusion

Ce travail pratique a permis d'appliquer de manière rigoureuse la méthode des éléments finis à la résolution d'un problème elliptique bidimensionnel. Les différentes étapes de la méthode — discrétisation du domaine, construction des fonctions de forme, assemblage de la matrice de rigidité, imposition des conditions aux limites et résolution du système linéaire — ont été mises en œuvre de façon cohérente et détaillée.

Les résultats numériques obtenus pour différents raffinements de maillage sont en accord avec la théorie et mettent en évidence la convergence de la solution lorsque le pas de discrétisation diminue. L'analyse de l'erreur montre cependant que la précision globale est limitée par la présence de singularités de gradient aux coins du domaine, induites par la discontinuité des conditions aux limites.

Ce travail constitue ainsi une base solide pour l'étude de problèmes elliptiques plus généraux, incluant des termes sources, des coefficients variables ou des stratégies de raffinement adaptatif visant à améliorer la précision au voisinage des zones singulières.

Annexes

```

1
2 ! =====
3 !   TP ELEMENTS FINIS
4 !   R solution 2D de l' equation de la chaleur (Laplace)
5 !   Methode des elements finis Q1 - SeaTech MOCA
6 ! =====
7
8 PROGRAM ELEMFINI
9
10 implicit none
11 integer, parameter :: dp = kind(1.0d0)
12
13 ! -----
14 ! Declarations generales
15 ! -----
16 integer :: Nx, Ny, Nn, nPoints, nElems
17 integer :: i, j, k, m, node
18 integer :: n1, n2, n3, n4
19 real(dp) :: Lx, Ly, dx, dy, alpha
20 real(dp), allocatable :: x(:), y(:), u(:), f(:)
21 real(dp), allocatable :: K(:, :)
22 real(dp), dimension(4,4) :: Ke
23 integer, dimension(4) :: nodes
24
25 ! -----
26 ! Lecture des param tres
27 ! -----
28 write(*,*) '===== '
29 write(*,*) 'Resolution-Delta_u=0 (FEM_Q1)'
30 write(*,*) '===== '
31 write(*,*) 'Entrer Nx, Ny (nombre d elements): '
32 read(*,*) Nx, Ny
33 write(*,*) 'Entrer Lx, Ly (dimensions du domaine): '
34 read(*,*) Lx, Ly
35
36 ! -----
37 ! Initialisation
38 ! -----
39 alpha = 1.0d6

```



```

40  dx = Lx / Nx
41  dy = Ly / Ny
42  Nn = (Nx+1)*(Ny+1)
43
44  allocate(x(Nn), y(Nn), u(Nn), f(Nn), K(Nn,Nn))
45  x = 0.d0; y = 0.d0
46  u = 0.d0; f = 0.d0
47  K = 0.d0
48
49  ! =====
50  ! 1. Construction du maillage
51  ! =====
52  node = 0
53  do j = 0, Ny
54      do i = 0, Nx
55          node = node + 1
56          x(node) = i * dx
57          y(node) = j * dy
58      end do
59  end do
60
61  ! =====
62  ! 2. Matrice élémentaire (Q1 - carrés réguliers)
63  ! =====
64  Ke = (1.0d0/6.0d0) * reshape([ &
65      4.d0, -1.d0, -2.d0, -1.d0, &
66      -1.d0, 4.d0, -1.d0, -2.d0, &
67      -2.d0, -1.d0, 4.d0, -1.d0, &
68      -1.d0, -2.d0, -1.d0, 4.d0 ], [4,4])
69
70  ! =====
71  ! 3. Assemblage de la matrice globale
72  ! =====
73  do j = 0, Ny-1
74      do i = 0, Nx-1
75          n1 = j*(Nx+1) + i + 1
76          n2 = n1 + 1
77          n3 = n2 + (Nx+1)
78          n4 = n1 + (Nx+1)
79          nodes = (/ n1, n2, n3, n4 /)
80          call assemble(K, nodes, Ke)

```

```

81     end do
82 end do
83
84 ! =====
85 ! 4. Conditions aux limites (p nalisation)
86 ! =====
87 do node = 1, Nn
88     if (abs(y(node)-Ly) < 1.d-12) then
89         K(node,node) = K(node,node) + alpha
90         f(node) = f(node) + alpha*100.d0
91     else if ( abs(x(node))<1.d-12 .or. abs(y(node))<1.d-12 &
92         .or. abs(x(node)-Lx)<1.d-12 ) then
93         K(node,node) = K(node,node) + alpha
94         f(node) = f(node)
95     end if
96 end do
97
98 ! =====
99 ! 5. R solution du syst me lin aire
100 ! =====
101 call gauss_solve(K, f, u, Nn)
102
103 ! =====
104 ! 6. Export de la solution (Tecplot)
105 ! =====
106 nPoints = Nn
107 nElems  = Nx*Ny
108
109 open(12, file='solution_general.dat', status='replace')
110 write(12,'(A)') 'TITLE_="Laplace_FEM_Q1"'
111 write(12,'(A)') 'VARIABLES_="X","Y","U"'
112 write(12,'(A,I0,A,I0,A)') 'ZONE_T="Domaine",_N=',nPoints,'_E=',
    nElems,'_F=FEPOINT_ET=QUADRILATERAL'
113
114 do node = 1, nPoints
115     write(12,'(3ES20.8)') x(node), y(node), u(node)
116 end do
117
118 do j = 0, Ny-1
119     do i = 0, Nx-1
120         n1 = j*(Nx+1) + i + 1

```

```

121      n2 = n1 + 1
122      n3 = n2 + (Nx+1)
123      n4 = n1 + (Nx+1)
124      write(12, '(4I10)') n1, n2, n3, n4
125  end do
126 end do
127 close(12)
128
129 write(*,*) 'Solution exportee : solution_general.dat'
130
131 ! =====
132 ! 7. Calcul et export de l'erreur analytique
133 ! =====
134 call erreur(Nx, Ny, Lx, Ly, u)
135
136 contains
137
138 ! =====
139 ! Assemblage    lmentaire
140 ! =====
141 subroutine assemble(K, nodes, Ke)
142   implicit none
143   real(dp), intent(inout) :: K(:, :)
144   real(dp), intent(in) :: Ke(4,4)
145   integer, intent(in) :: nodes(4)
146   integer :: i,j
147   do i = 1,4
148     do j = 1,4
149       K(nodes(i), nodes(j)) = K(nodes(i), nodes(j)) + Ke(i,j)
150     end do
151   end do
152 end subroutine assemble
153
154 ! =====
155 ! Solveur de Gauss
156 ! =====
157 subroutine gauss_solve(A,b,x,n)
158   implicit none
159   integer, intent(in) :: n
160   real(dp), intent(inout) :: A(n,n), b(n)
161   real(dp), intent(out) :: x(n)

```

```

162 integer :: i,j,k
163 real(dp) :: fact, somme
164
165 do k = 1, n-1
166     do i = k+1, n
167         fact = A(i,k)/A(k,k)
168         A(i,k:n) = A(i,k:n) - fact*A(k,k:n)
169         b(i) = b(i) - fact*b(k)
170     end do
171 end do
172
173 x(n) = b(n)/A(n,n)
174 do i = n-1, 1, -1
175     somme = 0.d0
176     do j = i+1, n
177         somme = somme + A(i,j)*x(j)
178     end do
179     x(i) = (b(i)-somme)/A(i,i)
180 end do
181 end subroutine gauss_solve
182
183 ! =====
184 ! Erreur analytique (s rie de Fourier)
185 ! =====
186 subroutine erreur(Nx, Ny, Lx, Ly, T)
187     implicit none
188     integer, intent(in) :: Nx, Ny
189     real(dp), intent(in) :: Lx, Ly
190     real(dp), intent(in) :: T(:)
191
192     integer :: i,j,k,m
193     real(dp) :: hx, hy, xi, yj, Tana, terme, somme
194     real(dp) :: pi, err
195
196     pi = 3.141592653589d0
197     hx = Lx / Nx
198     hy = Ly / Ny
199     k = 1
200
201     open(13, file='erreur.dat', status='replace')
202     write(13, '(A)') 'TITLE="Erreur_FEM_vs_analytique"'

```

```

203 write(13, '(A)') 'VARIABLES_="X","Y","U","err"'
204 write(13, '(A,IO,A,IO,A)') 'ZONE_T="Domaine",_N=',(Nx+1)*(Ny
    +1),'_E=',Nx*Ny,'_F=FEPOINT_ET=QUADRILATERAL'
205
206 do j = 0, Ny
207     yj = j*hy
208     do i = 0, Nx
209         xi = i*hx
210         if (i==0 .or. j==0 .or. i==Nx .or. j==Ny) then
211             if (j==Ny) then
212                 Tana = 100.d0
213             else
214                 Tana = 0.d0
215             end if
216         else
217             somme = 0.d0
218             do m = 0, 200
219                 terme = sin((2*m+1)*pi*xi/Lx)
220                 terme = terme * sinh((2*m+1)*pi*yj/Lx) / sinh((2*
                    m+1)*pi*Ly/Lx)
221                 somme = somme + terme/(2*m+1)
222             end do
223             Tana = 4.d0*100.d0/pi * somme
224         end if
225         err = abs(T(k)-Tana)
226         write(13, '(4E13.5)') xi, yj, T(k), err
227         k = k + 1
228     end do
229 end do
230 close(13)
231
232 write(*,*) 'Erreur_exportee:_erreur.dat'
233 end subroutine erreur
234
235 end program ELEMFINI

```

Listing 1 – Programme TP ELEMENTS FINIS – Résolution 2D de l'équation de la chaleur